Project 4 Help File

PART I

Suppose that we want to view a particular region in the plane which is delimited by some number of curves. In particular, suppose that we are interested in the region in the first quadrant ($x \geqslant 0$ and $y \geqslant 0$) bounded by the curves $x \leqslant y \leqslant 4x - x^2$. We can use the RegionPlot command to get a good idea what this region looks like. A little work shows us that that the curves $y = x$ and $y = 4x - x^2$ intersect at the origin and (3, 3).

In[∘]:=
```
RegionPlot[x ≤ y ≤ 4 x - x^2, {x, 0, 3}, {y, 0, 4}, FrameLabel → {x, y}, PlotPoints → 100]
```

Out[∘]=



Notice that the range of y was chosen so that we can see the entire region (you may need to do a little trial and error to get all of your given region). Hopefully the FrameLabel option is self-explanatory. The PlotPoints option ensures that *Mathematica* plots enough points to give us a decent view of the region in question. If your region has a fairly sharp corner, the default setting for PlotPoints may not give you a good picture.

If we want to integrate the function $f(x, y) = x^2 y^3 + 2xy + 3$ over this region, it is easiest to integrate in the order *dy dx* since the top and bottom curves are the same over the entire area.

In[∘]:=
```
Integrate[x^2*y^3 + 2 x*y+3, {x, 0 , 3}, {y, x, 4 x-x^2}]
```

Out[∘]= $\frac{575559}{1540}$

Note the order inside the integrate command. The first range, {*x, 0, 3*}, is the limits for the OUTER integral in your double integral. The second range, {*y, x, 4x - x^2*}, is the limits for the INNER integral. If you reverse these in the command, you will not get the right answer! In fact, you'll be left with a function of x (which is clearly the wrong answer).

If we wanted, we could have integrated in the order *dx dy*, but this involves breaking the integral into two pieces: $0 \le y \le 3$ and $3 \le y \le 4$. Note that $y = 4$ is the y-coordinate of the vertex of the parabola (which is why it is the upper limit of the second interval). In order to do the integral in this order, we need to solve $y = 4x - x^2$ for x. We could do this by completing the square. Or, we can let *Mathematica* do the work for us!

In[∘]:=
```
Solve[y == 4 x - x^2, x]
```

Out[∘]= $\{\{x \to 2 - \sqrt{4-y}\}, \{x \to 2 + \sqrt{4-y}\}\}$

Note that $x = 2 - \sqrt{4-y}$ is the half of the parabola to the left of the vertex while $x = 2 + \sqrt{4-y}$ is the portion to the right. So, on the interval $0 \le y \le 3$, we see that x is in the range $2 - \sqrt{4-y} \leqslant x \leqslant y$. Over $3 \le y \le 4$, x takes on the values $2 - \sqrt{4-y} \leqslant x \leqslant 2 + \sqrt{4-y}$. So, we should get the same answer by adding the two integral of f = f(*x,y*) over these regions.

In[∘]:=
```
Integrate[x^2*y^3 + 2 x*y+3, {y, 0 , 3}, {x, 2-Sqrt[4-y], y}] +
  Integrate[x^2*y^3 + 2 x*y+3, {y, 3 , 4}, {x, 2-Sqrt[4-y], 2+Sqrt[4-y]}]
```

Out[∘]= $\frac{575559}{1540}$

Part II

Suppose that we are interested in the polar region enclosed by $r = 2(1 + \cos(\theta))$ over the angle $0 \le \theta \leqslant 2\pi$. We can plot this region using the PolarPlot Command.

In[∘]:= `PolarPlot[2 (1 + Cos[θ]), {θ, 0, 2 * Pi}, AxesLabel → {x, y}, PlotStyle → Blue]`

Out[∘]=


This shape is known as a cardioid due to its similarity to a standard drawing of a heart. We could have also gotten a shaded version of this region using RegionPlot if we convert the polar coordinates back to Cartesian. It's easier to do this if we multiply through by r.

$r = 2\,(1 + \cos(\theta))$

$r^2 = 2\,(r + r\cos(\theta))$

$x^2 + y^2 = 2(\sqrt{x^2 + y^2} + x)$

In[∘]:=
```
RegionPlot[0 ≤ x^2 + y^2 ≤ 2 (Sqrt[x^2 + y^2] + x),
 {x, -1, 5}, {y, -3, 3}, FrameLabel → {x, y}, PlotPoints → 100]
```

Out[∘]=



If we want to integrate f(x,y) = $5x^2 - 3xy + 4y^2$ over this region, we first have to convert the function to polar.

f(x,y) = $5x^2 - 3xy + 4y^2$

f(r,θ) = $5r^2\cos^2\theta - 3r^2\sin\theta\cos\theta + 4r^2\sin^2\theta$

f(r,θ) = $4r^2 + r^2\cos^2\theta - 3r^2\sin\theta\cos\theta$

When you set up the integral, remember that $dx\,dy = r\,dr\,d\theta$. *Mathematica* WILL NOT automatically supply the extra factor of r. You must put it in manually.

In[∘]:=
```
Integrate[(4 r^2 + r^2 * Cos[θ]^2 - 3 r^2 * Sin[θ] * Cos[θ]) * r,
 {θ, 0, 2 * Pi}, {r, 0, 2 (1 + Cos[θ])}]
```

Out[∘]= $\frac{329\pi}{2}$